
An ExBuffer is a process that maintains a collection of items and flushes them once certain conditions have
been met.

  
    
  
  Installation


This package can be installed by adding :ex_buffer to your list of dependencies in mix.exs:
def deps do
  [
    {:ex_buffer "~> 0.3.0"}
  ]
end

  
    
  
  Documentation


For additional documentation, see HexDocs.

  
    
  
  Getting Started


We can easily start an ExBuffer by adding it directly to a supervision tree.
opts = [
  flush_callback: fn data, _ -> IO.inspect(data) end,
  max_length: 3,
  name: :buffer
]

children = [
  {ExBuffer, opts}
]

Supervisor.start_link(children, strategy: :one_for_one)
Once the buffer has been started, we can insert items.
ExBuffer.insert(:buffer, "foo")
ExBuffer.insert(:buffer, "bar")
And, once any of the configured conditions have been met, the buffer will automatically flush.
ExBuffer.insert(:buffer, "baz")
# ExBuffer flushes asynchronously and outputs ["foo", "bar", "baz"]

  
    
  
  Examples


ExBuffer is designed to be highly customizable, allowing it to be used in any number of scenarios.

  
    
  
  Simple Buffer


For example, we can use the ExBuffer behaviour to create a buffer with both a size limit and a time limit.
defmodule Buffer do
  use ExBuffer

  def start_link(opts \\ []) do
    opts = Keyword.merge([max_size: 8, buffer_timeout: 30_000], opts)
    ExBuffer.start_link(__MODULE__, opts)
  end

  def insert(item) do
    ExBuffer.insert(__MODULE__, item)
  end

  @impl ExBuffer
  def handle_flush(data, _opts) do
    IO.inspect(data)
  end

  @impl ExBuffer
  def handle_size(item) do
    byte_size(item) + 1
  end
end
We can easily start the Buffer process from above to see it in action.
Buffer.start_link()

Buffer.insert("foo")
Buffer.insert("bar")
# Buffer flushes asynchronously and outputs ["foo", "bar"]

Buffer.insert("baz")
# After 30 seconds pass...
# Buffer flushes asynchronously and outputs ["baz"]

  
    
  
  Partitioned Buffer


Alternatively, we could also use ExBuffer in conjunction with Elixir's PartitionSupervisor to easily
create a partitioned buffer with dynamic flush behavior.
defmodule PartitionedBuffer do
  use Supervisor

  def start_link(opts \\ []) do
    opts = Keyword.merge([max_length: 3], opts)
    Supervisor.start_link(__MODULE__, opts, name: __MODULE__)
  end

  def insert(item, partition) do
    ExBuffer.insert({:via, PartitionSupervisor, {:buffer, partition}}, item)
  end

  @impl Supervisor
  def init(opts) do
    part_sup_opts = [
      name: :buffer,
      child_spec: {ExBuffer, opts},
      partitions: 2,
      with_arguments: fn [opts], part ->
        flush_callback = fn data, _ -> handle_flush(data, part) end
        [Keyword.put(opts, :flush_callback, flush_callback)]
      end
    ]

    children = [
      {PartitionSupervisor, part_sup_opts}
    ]

    Supervisor.init(children, strategy: :one_for_one)
  end

  defp handle_flush(data, partition) do
    IO.inspect({partition, data})
  end
end
We can easily start the PartitionedBuffer process from above to see it in action.
PartitionedBuffer.start_link()

PartitionedBuffer.insert("foo", 0)
PartitionedBuffer.insert("foo", 1)
PartitionedBuffer.insert("bar", 0)
PartitionedBuffer.insert("bar", 1)
PartitionedBuffer.insert("baz", 0)
# Partition 0 flushes asynchronously and outputs {0, ["foo", "bar", "baz"]}

PartitionedBuffer.insert("baz", 1)
# Partition 1 flushes asynchronously and outputs {1, ["foo", "bar", "baz"]}



  

    
Change Log
    

All notable changes to this project will be documented in this file.

  
    
  
  [0.3.0] - 2023-10-26



  
    
  
  Added


	Added an ExBuffer behaviour.
	Added a non-banged version of ExBuffer.chunk!/2.


  
    
  
  Changed


	Split code back into multiple modules to simplify feature addition.
	Reorganized unit tests.


  
    
  
  Links


	GitHub


  
    
  
  [0.2.1] - 2023-10-24



  
    
  
  Changed


	Condensed code into a single module.


  
    
  
  Links


	GitHub
	HexDocs


  
    
  
  [0.2.0] - 2023-10-23



  
    
  
  Added


	Added the ability to provide a custom callback for determining item size (see :size_callback option for
both ExBuffer.start_link/1 and ExBuffer.chunk!/2).
	Added the ability to include metadata in the flush callback (see :flush_meta option for
ExBuffer.start_link/1).
	Added the ability to get the time until the next scheduled flush (see ExBuffer.next_flush/1).
	Added benchee dependency for local benchmarking.


  
    
  
  Changed


	The :callback option for ExBuffer.start_link/1 had been renamed to :flush_callback and now expects a
function of arity 2 (see ExBuffer.start_link/1).
	Updated function docs to include doctests.
	Updated README.md to include a more sophisticated use case.
	Updated README.md to display build status and Hex version.


  
    
  
  Links


	GitHub
	HexDocs


  
    
  
  [0.1.0] - 2023-10-20



  
    
  
  Added


	Initial release.


  
    
  
  Links


	HexDocs




  

    
ExBuffer behaviour
    



      
An ExBuffer is a process that maintains a collection of items and flushes
them once certain conditions have been met.
An ExBuffer can flush based on a timeout, a maximum length (item count), a
maximum byte size, or a combination of the three. When multiple conditions are
used, the ExBuffer will flush when the first condition is met.
ExBuffer also includes a number of helpful tools for testing and debugging.

      


      
        Summary


  
    Types
  


    
      
        error()

      


        Errors returned by ExBuffer functions.



    





  
    Callbacks
  


    
      
        handle_flush(data, opts)

      


        Invoked to flush an ExBuffer.



    


    
      
        handle_size(item)

      


        Invoked to determine the size of an inserted item.



    





  
    Functions
  


    
      
        chunk(enum, opts \\ [])

      


        Lazily chunks an enumerable based on ExBuffer flush conditions.



    


    
      
        chunk!(enum, opts \\ [])

      


        Lazily chunks an enumerable based on ExBuffer flush conditions and raises an ArgumentError
with invalid options.



    


    
      
        dump(buffer)

      


        Dumps the contents of the given ExBuffer to a list, bypassing a flush
callback and resetting the buffer.



    


    
      
        flush(buffer, opts \\ [])

      


        Flushes the given ExBuffer, regardless of whether or not the flush conditions
have been met.



    


    
      
        insert(buffer, item)

      


        Inserts the given item into the given ExBuffer.



    


    
      
        length(buffer)

      


        Returns the length (item count) of the given ExBuffer.



    


    
      
        next_flush(buffer)

      


        Returns the time (in ms) before the next scheduled flush of the given ExBuffer.



    


    
      
        size(buffer)

      


        Retuns the size (in bytes) of the given ExBuffer.



    


    
      
        start_link(module \\ nil, opts)

      


        Starts an ExBuffer process linked to the current process.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    error()


      
       
       View Source
     


  


  

      

          @type error() :: :invalid_callback | :invalid_limit


      


Errors returned by ExBuffer functions.

  


        

      

      
        Callbacks

        


  
    
      
      Link to this callback
    
    handle_flush(data, opts)


      
       
       View Source
     


  


  

      

          @callback handle_flush(data :: list(), opts :: keyword()) :: term()


      


Invoked to flush an ExBuffer.
The first argument (data) is a list of items inserted into the ExBuffer and the
second argument (opts) is a keyword list of flush options. See the :flush_callback
and :flush_meta options for ExBuffer.start_link/2 for more information.
This callback can return any term as the return value is disregarded by the ExBuffer.
This callback is required.

  



  
    
      
      Link to this callback
    
    handle_size(item)


      
       
       View Source
     


      (optional)

  


  

      

          @callback handle_size(item :: term()) :: non_neg_integer()


      


Invoked to determine the size of an inserted item.
The only argument (item) is any term that was inserted into the ExBuffer.
This callback must return a non-negative integer representing the item's byte size.
This callback is optional. See the :size_callback option for ExBuffer.start_link/2
for information about the default implementation.

  


        

      

      
        Functions

        


    

  
    
      
      Link to this function
    
    chunk(enum, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec chunk(
  Enumerable.t(),
  keyword()
) :: {:ok, Enumerable.t()} | {:error, error()}


      


Lazily chunks an enumerable based on ExBuffer flush conditions.
This function currently supports length and size conditions. If multiple
conditions are specified, a chunk is emitted once the first condition is
met (just like an ExBuffer process).
While this function is useful in it's own right, it's included primarily as
another way to synchronously test applications that use ExBuffer.

  
    
  
  Options


An enumerable can be chunked with the following options:
	:max_length - A non-negative integer representing the maximum allowed
length (item count) of a chunk. By default, there is no max length. (Optional)

	:max_size - A non-negative integer representing the maximum allowed size
(in bytes) of a chunk. The :size_callback option determines how item size
is computed. By default, there is no max size. (Optional)

	:size_callback - The function that will be invoked to deterime the size
of an item. For more information, see ExBuffer.start_link/2. (Optional)


Warning
Including neither :max_length nor :max_size is permitted but will result in a
single chunk being emitted. One can achieve a similar result in a more performant
way using Stream.into/2. In that same vein, including only a :max_length
condition makes this function a less performant version of Stream.chunk_every/2.
This function is optimized for chunking by either size or size and count. Any other
chunking strategy can likely be achieved in a more efficient way using other methods.


  
    
  
  Examples


iex> enum = ["foo", "bar", "baz", "foobar", "barbaz", "foobarbaz"]
...> {:ok, enum} = ExBuffer.chunk(enum, max_length: 3, max_size: 10)
...> Enum.into(enum, [])
[["foo", "bar", "baz"], ["foobar", "barbaz"], ["foobarbaz"]]

iex> enum = ["foo", "bar", "baz"]
...> {:ok, enum} = ExBuffer.chunk(enum, max_size: 8, size_callback: &(byte_size(&1) + 1))
...> Enum.into(enum, [])
[["foo", "bar"], ["baz"]]

iex> ExBuffer.chunk(["foo", "bar", "baz"], max_length: -5)
{:error, :invalid_limit}

  



    

  
    
      
      Link to this function
    
    chunk!(enum, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec chunk!(
  Enumerable.t(),
  keyword()
) :: Enumerable.t()


      


Lazily chunks an enumerable based on ExBuffer flush conditions and raises an ArgumentError
with invalid options.
For more information on this function's usage, purpose, and options, see ExBuffer.chunk!/2.

  
    
  
  Examples


iex> ["foo", "bar", "baz", "foobar", "barbaz", "foobarbaz"]
...> |> ExBuffer.chunk!(max_length: 3, max_size: 10)
...> |> Enum.into([])
[["foo", "bar", "baz"], ["foobar", "barbaz"], ["foobarbaz"]]

iex> ["foo", "bar", "baz"]
...> |> ExBuffer.chunk!(max_size: 8, size_callback: &(byte_size(&1) + 1))
...> |> Enum.into([])
[["foo", "bar"], ["baz"]]

iex> ExBuffer.chunk!(["foo", "bar", "baz"], max_length: -5)
** (ArgumentError) invalid limit

  



  
    
      
      Link to this function
    
    dump(buffer)


      
       
       View Source
     


  


  

      

          @spec dump(GenServer.server()) :: list()


      


Dumps the contents of the given ExBuffer to a list, bypassing a flush
callback and resetting the buffer.
While this functionality may occasionally be desriable in a production environment,
it is intended to be used primarily for testing and debugging.

  
    
  
  Example


iex> ExBuffer.insert(:buffer, "foo")
iex> ExBuffer.insert(:buffer, "bar")
iex> ExBuffer.dump(:buffer)
["foo", "bar"]

  



    

  
    
      
      Link to this function
    
    flush(buffer, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec flush(
  GenServer.server(),
  keyword()
) :: :ok


      


Flushes the given ExBuffer, regardless of whether or not the flush conditions
have been met.
While this functionality may occasionally be desriable in a production environment,
it is intended to be used primarily for testing and debugging.

  
    
  
  Options


An ExBuffer can be flushed with the following options:
	:async - A boolean representing whether or not the flush will be asynchronous.
By default, this value is true. (Optional)


  
    
  
  Example


iex> ExBuffer.insert(:buffer, "foo")
iex> ExBuffer.insert(:buffer, "bar")
...>
...> # Invokes callback on ["foo", "bar"]
iex> ExBuffer.flush(:buffer)
:ok

  



  
    
      
      Link to this function
    
    insert(buffer, item)


      
       
       View Source
     


  


  

      

          @spec insert(GenServer.server(), term()) :: :ok


      


Inserts the given item into the given ExBuffer.

  
    
  
  Example


iex> ExBuffer.insert(:buffer, "foo")
:ok

  



  
    
      
      Link to this function
    
    length(buffer)


      
       
       View Source
     


  


  

      

          @spec length(GenServer.server()) :: non_neg_integer()


      


Returns the length (item count) of the given ExBuffer.
While this functionality may occasionally be desriable in a production environment,
it is intended to be used primarily for testing and debugging.

  
    
  
  Example


iex> ExBuffer.insert(:buffer, "foo")
iex> ExBuffer.insert(:buffer, "bar")
iex> ExBuffer.length(:buffer)
2

  



  
    
      
      Link to this function
    
    next_flush(buffer)


      
       
       View Source
     


  


  

      

          @spec next_flush(GenServer.server()) :: non_neg_integer() | nil


      


Returns the time (in ms) before the next scheduled flush of the given ExBuffer.
If the given ExBuffer does not have a timeout, this function returns nil.
While this functionality may occasionally be desriable in a production environment,
it is intended to be used primarily for testing and debugging.

  
    
  
  Example


iex> next_flush = ExBuffer.next_flush(:buffer)
...>
...> # Assuming :buffer has a timeout...
iex> is_integer(next_flush)
true

  



  
    
      
      Link to this function
    
    size(buffer)


      
       
       View Source
     


  


  

      

          @spec size(GenServer.server()) :: non_neg_integer()


      


Retuns the size (in bytes) of the given ExBuffer.
For more information on how item size is computed, see ExBuffer.start_link/2.
While this functionality may occasionally be desriable in a production environment,
it is intended to be used primarily for testing and debugging.

  
    
  
  Example


iex> ExBuffer.insert(:buffer, "foo")
iex> ExBuffer.insert(:buffer, "bar")
iex> ExBuffer.size(:buffer)
6

  



    

  
    
      
      Link to this function
    
    start_link(module \\ nil, opts)


      
       
       View Source
     


  


  

      

          @spec start_link(
  module() | nil,
  keyword()
) :: GenServer.on_start()


      


Starts an ExBuffer process linked to the current process.
The first argument argument (module) is optional. It is intended to be used when
calling this function from a module that implements the ExBuffer behaviour. When
a module is passed, it may interact with the options that were passed in:
	If the module implements the handle_flush/2 callback, it will override the
:flush_callback option.

	If the module implements the handle_size/1 callback, it will override the
:size_callback option.

	If a :name option is not present, the module name will be used.



  
    
  
  Options


An ExBuffer can be started with the following options:
	:flush_callback - The function that will be invoked to handle a flush.
This function should expect two parameters: a list of items and a keyword
list of flush opts. The flush opts include the size and length of the buffer
at the time of the flush and optionally include any provided metadata (see
:flush_meta for more information). This function can return any term as the
return value is not used by the ExBuffer. (Required)

	:buffer_timeout - A non-negative integer representing the maximum time
(in ms) allowed between flushes of the ExBuffer. Once this amount of time
has passed, the ExBuffer will be flushed. By default, an ExBuffer does not
have a timeout. (Optional)

	:flush_meta - A term to be included in the flush opts under the meta key.
By default, this value will be nil. (Optional)

	:max_length - A non-negative integer representing the maximum allowed
length (item count) of the ExBuffer. Once the limit is hit, the ExBuffer will
be flushed. By default, an ExBuffer does not have a max length. (Optional)

	:max_size - A non-negative integer representing the maximum allowed size
(in bytes) of the ExBuffer. Once the limit is hit (or exceeded), the ExBuffer
will be flushed. The :size_callback option determines how item size is
computed. By default, an ExBuffer does not have a max size. (Optional)

	:size_callback - The function that will be invoked to determine the size
of an item. This function should expect a single parameter representing an
item and should return a single non-negative integer representing that item's
byte size. The default ExBuffer size callback is Kernel.byte_size/1
(:erlang.term_to_binary/1 is used to convert non-bitstring inputs to binary
if necessary). (Optional)


Additionally, an ExBuffer can also be started with any GenServer options.

